

NAG Fortran Library Routine Document

G13BJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13BJF produces forecasts of a time series (the output series) which depends on one or more other (input) series via a previously estimated multi-input model for which the state set information is not available. The future values of the input series must be supplied. In contrast with G13BHF the original past values of the input and output series are required. Standard errors of the forecasts are produced. If future values of some of the input series have been obtained as forecasts using ARIMA models for those series, this may be allowed for in the calculation of the standard errors.

2 Specification

```

SUBROUTINE G13BJF(MR, NSER, MT, PARA, NPARA, KFC, NEV, NFV, XXY, IXXY,
1          KZEF, RMSXY, MRX, PARX, IPARX, FVA, FSD, STTF, ISTTF,
2          NSTTF, WA, IWA, MWA, IMWA, IFAIL)
    INTEGER      MR(7), NSER, MT(4,NSER), NPARA, KFC, NEV, NFV, IXXY,
1          KZEF, MRX(7,NSER), IPARX, ISTTF, NSTTF, IWA,
2          MWA(IMWA), IMWA, IFAIL
    real
1          PARA(NPARA), XXY(IXXY,NSER), RMSXY(NSER),
2          PARX(IPARX,NSER), FVA(NFV), FSD(NFV), STTF(ISTTF),
          WA(IWA)

```

3 Description

The routine has two stages. The first stage is essentially the same as a call to the model estimation routine G13BEF, with zero iterations. In particular, all the parameters remain unchanged in the supplied input series transfer function models and output noise series ARIMA model. The internal nuisance parameters associated with the pre-observation period effects of the input series are estimated where requested, and so are any backforecasts of the output noise series. The output components z_t and n_t , and residuals a_t are calculated exactly as in Section 3 of the document for G13BEF, and the state set for forecasting is constituted.

The second stage is essentially the same as a call to the forecasting routine G13BHF. The same information is required, and the same information is returned.

Use of G13BJF should be confined to situations in which the state set for forecasting is unknown. Forecasting from the original data is relatively expensive because it requires recalculation of the state set. G13BJF returns the state set for use in producing further forecasts using G13BHF, or for updating the state set using G13BGF.

4 References

Box G E P and Jenkins G M (1976) *Time Series Analysis: Forecasting and Control* (Revised Edition) Holden-Day

5 Parameters

- 1: MR(7) – INTEGER array *Input/Output*
On entry: the orders vector (p, d, q, P, D, Q, s) , of the ARIMA model for the output noise component.

p , q , P , Q refer respectively to the number of autoregressive (ϕ), moving average (θ), seasonal autoregressive (Φ) and seasonal moving average (Θ) parameters.

d , D , s refer respectively to the order of non-seasonal differencing, the order of seasonal differencing and the seasonal period.

On exit: used as internal workspace prior to being restored and hence is unchanged.

2: NSER – INTEGER *Input*

On entry: the number of input and output series. There may be any number of input series (including none), but only one output series.

3: MT(4,NSER) – INTEGER array *Input*

On entry: the transfer function model orders b , p and q of each of the input series. The data for input series i is held in column i . Row 1 holds the value b_i , row 2 holds the value q_i and row 3 holds the value p_i .

For a simple input, $b_i = q_i = p_i = 0$.

Row 4 holds the value r_i , where $r_i = 1$ for a simple input, and $r_i = 2$ or 3 for a transfer function input.

The choice $r_i = 3$ leads to estimation of the pre-period input effects as nuisance parameters, and $r_i = 2$ suppresses this estimation. This choice may affect the returned forecasts and the state set.

When $r_i = 1$, any non-zero contents of rows 1,2 and 3 of column i are ignored.

Constraint: $MT(4,i) = 1, 2 \text{ or } 3$, for $i = 1, 2, \dots, NSER - 1$.

4: PARA(NPARA) – *real* array *Input/Output*

On entry: estimates of the multi-input model parameters. These are in order, firstly the ARIMA model parameters: p values of ϕ parameters, q values of θ parameters, P values of Φ parameters, Q values of Θ parameters.

These are followed by the transfer function model parameter values $\omega_0, \omega_1, \dots, \omega_{q_1}, \delta_1, \dots, \delta_{p_1}$ for the first of any input series and similarly for each subsequent input series. The final component of PARA is the value of the constant c .

On exit: the parameter values may be updated using an additional iteration in the estimation process.

5: NPARA – INTEGER *Input*

On entry: the exact number of ϕ , θ , Φ , Θ , ω , δ , c parameters, so that $NPARA = p + q + P + Q + NSER + \sum(p_i + q_i)$, the summation being over all the input series. (c must be included whether its value was previously estimated or was set fixed.)

6: KFC – INTEGER *Input*

On entry: KFC must be set to 1 if the constant was estimated when the model was fitted, and 0 if it was held at a fixed value. This only affects the degrees of freedom used in calculating the estimated residual variance.

Constraint: $KFC = 0 \text{ or } 1$.

7: NEV – INTEGER *Input*

On entry: the number of original (undifferenced) values in each of the input and output time series.

8: NFV – INTEGER *Input*

On entry: the number of forecast values of the output series required.

Constraint: $NFV > 0$.

- 9: XXY(IXXY,NSER) – *real* array Input/Output
- On entry:* the columns of XXY must contain in the first NEV places, the past values of each of the input and output series, in that order. In the next NFV places, the columns relating to the input series (i.e., columns 1 to NSER – 1) contain the future values of the input series which are necessary for construction of the forecasts of the output series y .
- On exit:* if KZEF = 0 then XXY is unchanged except that the relevant NFV values in the column relating to the output series (column NSER) contain the forecast values (FVA), but if KZEF \neq 0 then the columns of XXY contain the corresponding values of the input component series z_t and the values of the output noise component n_t , in that order.
- 10: IXXY – INTEGER Input
- On entry:* the first dimension of the array XXY as declared in the (sub)program from which G13BJF is called.
- Constraint:* IXXY \geq (NEV + NFV).
- 11: KZEF – INTEGER Input
- On entry:* KZEF must be set to 0 if the relevant NFV values of the forecasts (FVA) are to be held in the output series column (NSER) of XXY (which is otherwise unchanged) on exit, and must not be set to 0 if the values of the input component series z_t and the values of the output noise component n_t are to overwrite the contents of XXY on exit.
- 12: RMSXY(NSER) – *real* array Input/Output
- On entry:* the first (NSER – 1) elements of RMSXY must contain the estimated residual variance of the input series ARIMA models. In the case of those inputs for which no ARIMA model is available or its effects are to be excluded in the calculation of forecast standard errors, the corresponding entry of RMSXY should be set to 0.
- On exit:* RMSXY(NSER) contains the estimated residual variance of the output noise ARIMA model which is calculated from the supplied series. Otherwise RMSXY is unchanged.
- 13: MRX(7,NSER) – INTEGER array Input/Output
- On entry:* the orders array for each of the input series ARIMA models. Thus, column i contains values of p, d, q, P, D, Q, s for input series i . In the case of those inputs for which no ARIMA model is available, the corresponding orders should be set to 0.
- On exit:* unchanged, except for column NSER which is used as workspace.
- 14: PARX(IPARX,NSER) – *real* array Input
- On entry:* values of the parameters (ϕ, θ, Φ , and Θ) for each of the input series ARIMA models. Thus column i contains MRX(1, i) values of ϕ , MRX(3, i) values of θ , MRX(4, i) values of Φ and MRX(6, i) values of Θ , in that order.
- Values in the columns relating to those input series for which no ARIMA model is available are ignored.
- 15: IPARX – INTEGER Input
- On entry:* the first dimension of the array PARX as declared in the (sub)program from which G13BJF is called.
- Constraint:* IPARX \geq nce, where nce is the maximum number of parameters in any of the input series ARIMA models. If there are no input series, then IPARX \geq 1.
- 16: FVA(NFV) – *real* array Output
- On exit:* the required forecast values for the output series. (Note that these are also output in column NSER of XXY if KZEF = 0.)

- 17: FSD(NFV) – *real* array Output
On exit: the standard errors for each of the forecast values.
- 18: STTF(ISTTF) – *real* array Output
On exit: the NSTTF values of the state set based on the first NEV sets of (past) values of the input and output series.
- 19: ISTTF – INTEGER Input
On entry: the dimension of the array STTF as declared in the (sub)program from which G13BJF is called.
Constraint: $ISTTF \geq (p \times s) + d + (D \times s) + q + \max(p, Q \times s) + ncf$, where $ncf = \sum (b_i + q_i + p_i)$ and the summation is over all input series for which $r_i > 1$.
- 20: NSTTF – INTEGER Output
On exit: the number of values in the state set array STTF.
- 21: WA(IWA) – *real* array Workspace
 22: IWA – INTEGER Input
On entry: the dimension of the array WA as declared in the (sub)program from which G13BJF is called.

It is not practical to outline a method for deriving the exact minimum permissible value of IWA, but the following gives a reasonably good approximation which tends to be on the conservative side.

Note: there are three error indicators associated with IWA. These are IFAIL = 4, 5 and 6. The first of these probably indicates an abnormal entry value of NFV, while the second indicates that IWA is much too small and needs to be increased by a substantial amount. The last of these indicates that IWA is too small but returns the necessary minimum value in MWA(1).

Let $q' = q + (Q \times s)$ and $d' = d + (D \times s)$, where the output noise ARIMA model orders are p, d, q, P, D, Q, s .

Let there be l input series, where $l = \text{NSER} - 1$.

Let $mx_i = \max(b_i + q_i, p_i)$, if $r_i = 3$ for $i = 1, 2, \dots, l$ if $l > 0$
 $mx_i = 0$, if $r_i \neq 3$ for $i = 1, 2, \dots, l$ if $l > 0$

where the transfer function model orders of input series i are given by b_i, q_i, p_i, r_i .

Let $qx = \max(q', mx_1, mx_2, \dots, mx_l)$

Let $ncg = \text{NPARA} + qx + \sum_{i=1}^l mx_i$ and $nch = N + d + 6 \times qx$.

Finally, let $nci = \text{NSER}$, and then increment nci by 1 every time any of the following conditions are satisfied. (The last two conditions should be applied separately to each input series, so that for example if we have two input series and $p_1 > 0$ and $p_2 > 0$, then nci is incremented by 2 in respect of these.)

The conditions are:

$p > 0$
 $q > 0$
 $P > 0$
 $Q > 0$
 $qx > 0$
 $mx_i > 0$, separately for $i = 1, 2, \dots, l$, if $l > 0$
 $p_i > 0$, separately for $i = 1, 2, \dots, l$, if $l > 0$,

then $\text{IWA} > 2 \times (ncg)^2 + nch \times (nci + 4)$.

23: MWA(IMWA) – INTEGER array

Workspace

24: IMWA – INTEGER

Input

On entry: the dimension of the array MWA as declared in the (sub)program from which G13BJF is called.

Constraint: $IMWA \geq (16 \times NSER) + (7 \times ncg) + (3 \times NPARA) + 27$, where the derivation of ncg is described under IWA above.

When IMWA is too small, as indicated by IFAIL = 7, the requisite minimum value of IMWA is returned in MWA(1)..

25: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, KFC < 0,
or KFC > 1,
or IXXY < (NEV + NFV),
or NFV ≤ 0.

IFAIL = 2

On entry, IPARX is too small or NPARA is inconsistent with the orders specified in arrays MR and MT; or one of the r_i , stored in MT(4, i), does not equal 1, 2 or 3.

IFAIL = 3

On entry or during execution, one or more sets of δ parameters do not satisfy the stationarity or invertibility test conditions.

IFAIL = 4

On entry, IWA is too small for the final forecasting calculations. This is a highly unlikely error, and would probably indicate that NFV was abnormally large.

IFAIL = 5

On entry, IWA is too small by a very considerable margin. No information is supplied about the requisite minimum size.

IFAIL = 6

On entry, IWA is too small, but the requisite minimum size is returned in MWA(1).

IFAIL = 7

On entry, IMWA is too small, but the requisite minimum size is returned in MWA(1).

IFAIL = 8

This indicates a failure in F04ASF, which is used to solve the equations giving the latest estimates of the parameters.

IFAIL = 9

This indicates a failure in the inversion of the second derivative matrix associated with parameter estimation.

IFAIL = 10

On entry, or during execution, one or more sets of the ARIMA (ϕ , θ , Φ or Θ) parameters do not satisfy the stationarity or invertibility test conditions.

IFAIL = 11

On entry, ISTTF is too small.

7 Accuracy

The computations are believed to be stable.

8 Further Comments

The time taken by the routine is approximately proportional to the product of the length of each series and the square of the number of parameters in the multi-input model.

9 Example

The data in the example relates to 40 observations of an output time series and 5 input time series. This differs from the example in G13BEF in that there are now 4 simple input series. The output series has one autoregressive (ϕ) parameter and one seasonal moving average (Θ) parameter. The seasonal period is 4. The transfer function input (the fifth in the set) is defined by orders $b_5 = 1$, $q_5 = 0$, $p_5 = 1$, $r_5 = 3$, so that it allows for pre-observation period effects. The initial values of the specified model are:

$$\phi = 0.495, \Theta = 0.238, \omega_1 = -0.367 \quad \omega_2 = -3.876 \quad \omega_3 = 4.516$$

$$\omega_4 = 2.474 \quad \omega_{5,1} = 8.629 \quad \delta_{5,1} = 0.688, c = -82.858.$$

A further 8 values of the input series are supplied, and it is assumed that the values for the fifth series have themselves been forecast from an ARIMA model with orders 2 0 2 0 1 1 4, in which $\phi_1 = 1.6743$, $\phi_2 = -0.9505$, $\theta_1 = 1.4605$, $\theta_2 = -0.4862$ and $\Theta_1 = 0.8993$, and for which the residual mean square is 0.1720.

The following are computed and printed out: the state set after initial processing of the original 40 sets of values, the estimated residual variance for the output noise series, the 8 forecast values and their standard errors, and the values of the components z_t and the output noise component n_t .

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G13BJF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NSERMX, NPMAX, IPARX, NFVMAX, ISTTF, NEVMAX,
+                     IXXY, IWA, IMWA
      PARAMETER        (NSERMX=6, NPMAX=10, IPARX=8, NFVMAX=10, ISTTF=20,
+                     NEVMAX=40, IXXY=NEVMAX+NFVMAX, IWA=1500, IMWA=250)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5, NOUT=6)
*      .. Local Scalars ..
```

```

      INTEGER          I, IFAIL, J, KFC, KZEF, N, NEV, NFV, NPARA, NSER,
+      NSTTF
*      .. Local Arrays ..
      real             FSD(NFVMAX), FVA(NFVMAX), PARA(NFVMAX),
+      PARX(IPARX,NSERMX), RMSXY(NSERMX), STTF(ISTTF),
+      WA(IWA), XXY(IXXY,NSERMX)
      INTEGER          MR(7), MRX(7,NSERMX), MT(4,NSERMX), MWA(IMWA)
*      .. External Subroutines ..
      EXTERNAL         G13BJF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G13BJF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) KFC, NEV, NFV, NSER, KZEF
      IF (NSER.GT.0 .AND. NSER.LE.NSERMX .AND. NFV.GT.0 .AND. NFV.LE.
+      NFVMAX .AND. NEV.GT.0 .AND. NEV.LE.NEVMAX) THEN
        READ (NIN,*) (MR(I),I=1,7)
        DO 20 I = 1, 4
          READ (NIN,*) (MT(I,J),J=1,NSER)
20      CONTINUE
        NPARA = 0
        DO 40 I = 1, NSER
          NPARA = NPARA + MT(2,I) + MT(3,I)
40      CONTINUE
        NPARA = NPARA + MR(1) + MR(3) + MR(4) + MR(6) + NSER
        IF (NPARA.LE.NPMAX) THEN
          READ (NIN,*) (PARA(I),I=1,NPARA)
          N = NEV + NFV
          DO 60 I = 1, N
            READ (NIN,*) (XXY(I,J),J=1,NSER)
60      CONTINUE
          READ (NIN,*) (RMSXY(I),I=1,NSER)
          DO 80 I = 1, 7
            READ (NIN,*) (MRX(I,J),J=1,NSER)
80      CONTINUE
          DO 100 I = 1, 5
            READ (NIN,*) (PARX(I,J),J=1,NSER)
100     CONTINUE
          IFAIL = 1
*
          CALL G13BJF(MR,NSER,MT,PARA,NPARA,KFC,NEV,NFV,XXY,IXXY,KZEF,
+          RMSXY,MRX,PARX,IPARX,FVA,FSD,STTF,ISTTF,NSTTF,
+          WA,IWA,MWA,IMWA,IFAIL)
*
          WRITE (NOUT,*)
          IF (IFAIL.NE.0) WRITE (NOUT,99999) 'G13BJF fails. IFAIL = ',
+          IFAIL
          IF (IFAIL.EQ.0 .OR. IFAIL.EQ.8 .OR. IFAIL.EQ.9 .OR.
+          IFAIL.EQ.11) THEN
            WRITE (NOUT,99999) 'After processing', NEV,
+            ' sets of observations'
            WRITE (NOUT,99998) NSTTF,
+            ' values of the state set are derived'
            WRITE (NOUT,*)
            WRITE (NOUT,99997) (STTF(I),I=1,NSTTF)
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'The residual mean square for the output'
            WRITE (NOUT,99996)
+            'series is also derived and its value is', RMSXY(NSER)
            WRITE (NOUT,*)
            WRITE (NOUT,*)
+            'The forecast values and their standard errors are'
            WRITE (NOUT,*)
            WRITE (NOUT,*) '      I          FVA          FSD'
            WRITE (NOUT,*)
            DO 120 I = 1, NFV
              WRITE (NOUT,99995) I, FVA(I), FSD(I)
120     CONTINUE
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'The values of z(t) and n(t) are'
            WRITE (NOUT,*)

```

```

      +      '      I      z1      z2      z3      z4      z5      n'
      WRITE (NOUT,*)
      DO 140 I = 1, N
          WRITE (NOUT,99994) I, (XXY(I,J),J=1,NSER)
140      CONTINUE
      END IF
      END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,I3,A)
99998 FORMAT (1X,I3,A)
99997 FORMAT (1X,6F10.4)
99996 FORMAT (1X,A,F10.4)
99995 FORMAT (1X,I4,F10.3,F10.4)
99994 FORMAT (1X,I4,6F10.3)
      END

```

9.2 Program Data

G13BJF Example Program Data

1	40	8	6	1					
1	0	0	0	0	1	4			
0	0	0	0	1	0				
0	0	0	0	0	0				
0	0	0	0	1	0				
1	1	1	1	3	0				
0.4950	0.2380	-0.3670	-3.8760	4.5160	2.4740	8.6290	0.6880		
-82.8580									
1.0	1.0	0.0	0.0	8.075	105.0				
1.0	0.0	1.0	0.0	7.819	119.0				
1.0	0.0	0.0	1.0	7.366	119.0				
1.0	-1.0	-1.0	-1.0	8.113	109.0				
2.0	1.0	0.0	0.0	7.380	117.0				
2.0	0.0	1.0	0.0	7.134	135.0				
2.0	0.0	0.0	1.0	7.222	126.0				
2.0	-1.0	-1.0	-1.0	7.768	112.0				
3.0	1.0	0.0	0.0	7.386	116.0				
3.0	0.0	1.0	0.0	6.965	122.0				
3.0	0.0	0.0	1.0	6.478	115.0				
3.0	-1.0	-1.0	-1.0	8.105	115.0				
4.0	1.0	0.0	0.0	8.060	122.0				
4.0	0.0	1.0	0.0	7.684	138.0				
4.0	0.0	0.0	1.0	7.580	135.0				
4.0	-1.0	-1.0	-1.0	7.093	125.0				
5.0	1.0	0.0	0.0	6.129	115.0				
5.0	0.0	1.0	0.0	6.026	108.0				
5.0	0.0	0.0	1.0	6.679	100.0				
5.0	-1.0	-1.0	-1.0	7.414	96.0				
6.0	1.0	0.0	0.0	7.112	107.0				
6.0	0.0	1.0	0.0	7.762	115.0				
6.0	0.0	0.0	1.0	7.645	123.0				
6.0	-1.0	-1.0	-1.0	8.639	122.0				
7.0	1.0	0.0	0.0	7.667	128.0				
7.0	0.0	1.0	0.0	8.080	136.0				
7.0	0.0	0.0	1.0	6.678	140.0				
7.0	-1.0	-1.0	-1.0	6.739	122.0				
8.0	1.0	0.0	0.0	5.569	102.0				
8.0	0.0	1.0	0.0	5.049	103.0				
8.0	0.0	0.0	1.0	5.642	89.0				
8.0	-1.0	-1.0	-1.0	6.808	77.0				
9.0	1.0	0.0	0.0	6.636	89.0				
9.0	0.0	1.0	0.0	8.241	94.0				
9.0	0.0	0.0	1.0	7.968	104.0				
9.0	-1.0	-1.0	-1.0	8.044	108.0				
10.0	1.0	0.0	0.0	7.791	119.0				
10.0	0.0	1.0	0.0	7.024	126.0				
10.0	0.0	0.0	1.0	6.102	119.0				
10.0	-1.0	-1.0	-1.0	6.053	103.0				

11.0	1.0	0.0	0.0	5.941	0.0
11.0	0.0	1.0	0.0	5.386	0.0
11.0	0.0	0.0	1.0	5.811	0.0
11.0	-1.0	-1.0	-1.0	6.716	0.0
12.0	1.0	0.0	0.0	6.923	0.0
12.0	0.0	1.0	0.0	6.939	0.0
12.0	0.0	0.0	1.0	6.705	0.0
12.0	-1.0	-1.0	-1.0	6.914	0.0
0.0	0.0	0.0	0.0	0.1720	0.0
0	0	0	0	2	0
0	0	0	0	0	0
0	0	0	0	2	0
0	0	0	0	0	0
0	0	0	0	1	0
0	0	0	0	1	0
0	0	0	0	4	0
0.0	0.0	0.0	0.0	1.6743	0.0
0.0	0.0	0.0	0.0	-0.9505	0.0
0.0	0.0	0.0	0.0	1.4605	0.0
0.0	0.0	0.0	0.0	-0.4862	0.0
0.0	0.0	0.0	0.0	0.8993	0.0

9.3 Program Results

G13BJF Example Program Results

After processing 40 sets of observations
6 values of the state set are derived

6.0530 193.8741 2.0790 -2.8580 -3.5906 -2.5203

The residual mean square for the output
series is also derived and its value is 20.7599

The forecast values and their standard errors are

I	FVA	FSD
1	93.398	4.5563
2	96.958	6.2172
3	86.046	7.0933
4	77.589	7.3489
5	82.139	7.3941
6	96.276	7.5823
7	98.345	8.1445
8	93.577	8.8536

The values of $z(t)$ and $n(t)$ are

I	z1	z2	z3	z4	z5	n
1	-0.339	-3.889	0.000	0.000	188.603	-79.375
2	-0.339	-0.000	4.514	0.000	199.438	-84.613
3	-0.339	-0.000	0.000	2.479	204.683	-87.823
4	-0.339	3.889	-4.514	-2.479	204.383	-91.940
5	-0.678	-3.889	0.000	0.000	210.623	-89.056
6	-0.678	-0.000	4.514	0.000	208.591	-77.426
7	-0.678	-0.000	0.000	2.479	205.070	-80.870
8	-0.678	3.889	-4.514	-2.479	203.407	-87.624
9	-1.017	-3.889	0.000	0.000	206.974	-86.068
10	-1.017	-0.000	4.514	0.000	206.132	-87.628
11	-1.017	-0.000	0.000	2.479	201.920	-88.381
12	-1.017	3.889	-4.514	-2.479	194.819	-75.698
13	-1.356	-3.889	0.000	0.000	203.974	-76.729
14	-1.356	-0.000	4.514	0.000	209.884	-75.041
15	-1.356	-0.000	0.000	2.479	210.705	-76.828
16	-1.356	3.889	-4.514	-2.479	210.373	-80.912
17	-1.695	-3.889	0.000	0.000	205.942	-85.358
18	-1.695	-0.000	4.514	0.000	194.575	-89.394
19	-1.695	-0.000	0.000	2.479	185.866	-86.650

20	-1.695	3.889	-4.514	-2.479	185.509	-84.709
21	-2.035	-3.889	0.000	0.000	191.606	-78.682
22	-2.035	-0.000	4.514	0.000	193.194	-80.673
23	-2.035	-0.000	0.000	2.479	199.896	-77.340
24	-2.035	3.889	-4.514	-2.479	203.497	-76.358
25	-2.374	-3.889	0.000	0.000	214.552	-80.290
26	-2.374	-0.000	4.514	0.000	213.770	-79.910
27	-2.374	-0.000	0.000	2.479	216.796	-76.901
28	-2.374	3.889	-4.514	-2.479	206.780	-79.302
29	-2.713	-3.889	0.000	0.000	200.416	-91.814
30	-2.713	-0.000	4.514	0.000	185.941	-84.742
31	-2.713	-0.000	0.000	2.479	171.495	-82.261
32	-2.713	3.889	-4.514	-2.479	166.673	-83.857
33	-3.052	-3.889	0.000	0.000	173.418	-77.477
34	-3.052	-0.000	4.514	0.000	176.573	-84.035
35	-3.052	-0.000	0.000	2.479	192.594	-88.021
36	-3.052	3.889	-4.514	-2.479	201.261	-87.105
37	-3.391	-3.889	0.000	0.000	207.879	-81.599
38	-3.391	-0.000	4.514	0.000	210.249	-85.372
39	-3.391	-0.000	0.000	2.479	205.262	-85.350
40	-3.391	3.889	-4.514	-2.479	193.874	-84.379
41	-3.730	-3.889	0.000	0.000	185.617	-84.600
42	-3.730	0.000	4.514	0.000	178.969	-82.795
43	-3.730	0.000	0.000	2.479	169.607	-82.309
44	-3.730	3.889	-4.514	-2.479	166.832	-82.409
45	-4.069	-3.889	0.000	0.000	172.733	-82.636
46	-4.069	0.000	4.514	0.000	178.579	-82.748
47	-4.069	0.000	0.000	2.479	182.739	-82.804
48	-4.069	3.889	-4.514	-2.479	183.582	-82.831
